
ABSTRACT 
Developers are often challenged to build complex combinations of large files with a good performance. 

In most cases the SQL procedure is the obvious choice. However, the DATA step often provides better alternatives 
with outstanding performance improvements. 

An extra level of complexity is added when data can not just be combined on common key fields, but time slices or 
weighting parameters have to be taken into account. 

INTRODUCTION 
We will learn how to build solutions with a good performance for complex combinations of tables with millions of 
records. 

We will compare the following methods: 

• SQL inner joins 
• MERGE - BY statements 
• hash tables 
• arrays 
• multiple SET statements 

The following business cases will be discussed: 

• How can we efficiently treat data in a star schema with a fact table and several dimension tables? 
• How can we efficiently treat time-sliced data where every record must be split into several records using a weight 

variable from another table? 
• How can we efficiently combine two time-sliced tables and search for gaps in the combination of these tables? 

BUSINESS CASE 1 
Combining a Fact Table with Several Dimension Tables 

INPUT TABLES 
We have data in a star schema with a fact table (PORTFOLIO) and several dimension tables (CLIENT, 
INSTALLATION, CONTRACT and INVOICING) with unique key values. 

We will combine the data from the PORTFOLIO fact table with the CLIENT, INSTALLATION, CONTRACT and 
INVOICING dimension tables using the following methods: 

• SQL inner joins 
• DATA step MERGE - BY statements 
• hash tables 
 

The resulting table will look as follows: 

 

 
 

SOLUTION 1 
Using SQL Inner Joins 
Combine all tables at once in one query using SQL inner joins. 

 

PROC SQL;  
   CREATE TABLE BC1_SOL1 AS 
      SELECT POR.*, 
             CLI.CLIENT_NAME, 
             CON.SERVICE, 
             INS.POSTAL_CODE, 
             INV.ACCOUNT,  
             INV.DIRECT_DEBIT 
         FROM PORTFOLIO    POR, 
              CLIENT       CLI, 
              CONTRACT     CON, 
              INSTALLATION INS, 
              INVOICING    INV 
         WHERE POR.INSTALLATION = INS.INSTALLATION AND 
               POR.CONTRACT     = CON.CONTRACT     AND 
               POR.CLIENT       = CLI.CLIENT       AND 
               POR.CONTRACT     = INV.CONTRACT; 
QUIT; 

SOLUTION 2 
Using DATA Step MERGE – BY Statements 
Combine the PORTFOLIO fact table with the INSTALLATION dimension table by INSTALLATION using a first DATA 
step MERGE statement. 

 
PROC SORT DATA = PORTFOLIO; 
   BY INSTALLATION; 
RUN; 
 
PROC SORT DATA = INSTALLATION; 
   BY INSTALLATION; 
RUN; 
 
DATA PORTFOLIO_INSTALLATION; 
   MERGE PORTFOLIO (IN = POR) 
         INSTALLATION; 
   BY INSTALLATION; 
   IF POR; 
RUN; 

 

Combine the intermediate result with the CONTRACT and INVOICING dimension tables by CONTRACT using a 
second DATA step MERGE statement. 

 
PROC SORT DATA = PORTFOLIO_INSTALLATION; 
   BY CONTRACT; 
RUN; 
 
PROC SORT DATA = CONTRACT; 
   BY CONTRACT; 
RUN; 
 
PROC SORT DATA = INVOICING; 
   BY CONTRACT; 
RUN; 
 
DATA PORTFOLIO_INSTALLATION_CONTRACT; 
   MERGE PORTFOLIO_INSTALLATION (IN = POR) 
         CONTRACT 
         INVOICING; 
   BY CONTRACT; 
   IF POR; 
RUN; 

 

Combine the intermediate result with the CLIENT dimension table by CLIENT using a third DATA step MERGE 
statement. 

 
PROC SORT DATA = PORTFOLIO_INSTALLATION_CONTRACT; 
   BY CLIENT; 
RUN; 
 
PROC SORT DATA = CLIENT; 
   BY CLIENT; 
RUN; 
 
DATA BC1_SOL2; 
   MERGE PORTFOLIO_INSTALLATION_CONTRACT (IN = POR) 
         CLIENT; 
   BY CLIENT; 
   IF POR; 
RUN; 

SOLUTION 3 
Using Hash Tables 
Combine the PORTFOLIO fact table with all the dimension tables at once in one DATA step by loading the dimension 
tables into hash objects. 

 
DATA BC1_SOL3; 
   SET PORTFOLIO; 
   LENGTH CLIENT_NAME  $ 40 
          SERVICE      $  3 
          POSTAL_CODE  $  4 
          ACCOUNT      $  8 
          DIRECT_DEBIT $  1; 
   IF _N_ = 1 THEN DO; 
      DECLARE HASH CLI (DATASET : 'CLIENT');   
      CLI.DEFINEKEY  ('CLIENT'); 
      CLI.DEFINEDATA ('CLIENT_NAME'); 
      CLI.DEFINEDONE ( ); 
      DECLARE HASH CON (DATASET : 'CONTRACT');   
      CON.DEFINEKEY  ('CONTRACT'); 
      CON.DEFINEDATA ('SERVICE'); 
      CON.DEFINEDONE ( ); 
      DECLARE HASH INS (DATASET : 'INSTALLATION');   
      INS.DEFINEKEY  ('INSTALLATION'); 
      INS.DEFINEDATA ('POSTAL_CODE'); 
      INS.DEFINEDONE ( ); 
      DECLARE HASH INV (DATASET : 'INVOICING');   
      INV.DEFINEKEY  ('CONTRACT'); 
      INV.DEFINEDATA ('ACCOUNT', 'DIRECT_DEBIT'); 
      INV.DEFINEDONE ( ); 
   END; 
   DROP RC; 
   RC = CLI.FIND ( );   
   RC = CON.FIND ( ); 
   RC = INS.FIND ( ); 
   RC = INV.FIND ( ); 
RUN; 

 

 Load the CLIENT table into a hash object using the CLIENT variable as the key and the CLIENT_NAME values 
as the data. 

 Load the CONTRACT table into a hash object using the CONTRACT variable as the key and the SERVICE 
values as the data. 

 Load the INSTALLATION table into a hash object using the INSTALLATION variable as the key and the 
POSTAL_CODE values as the data. 

 Load the INVOICING table into a hash object using the CONTRACT variable as the key and the ACCOUNT and 
DIRECT_DEBIT values as the data. 

 Look up the dimension values in the hash objects. 

RESOURCE USAGE 
The following tables and charts illustrate resource usage in terms of: 

• real time 
• user CPU 
• memory 
 

 
 

 
 

 
 

Beyond the BasicsSAS Global Forum 2010

 



6 

BUSINESS CASE 2 
Splitting Time-sliced Data Using a Weight Variable 

INPUT TABLES 
The METERING table contains the CONSUMPTION measured for an INSTALLATION between 2 dates (MET_FROM 
and MET_TO). 

The SLP_DAY table contains the weight (SLP) for every day (DATE). 

We have to split the CONSUMPTION from the METERING table into records per year using the weight variable SLP 
from the SLP_DAY table. 

 

 
 

We will split the CONSUMPTION per year using the following steps: 

1. Make the daily SLP values available using one of the following methods: 

• SQL joins 
• a hash table 
• an array 

2. Calculate the total SLP value for the consumption period (TOTAL_SLP) using the following formula: 

 
 
3. Calculate the total yearly SLP values (YEAR_SLP) using the following formulas: 

 
 
4. Split the CONSUMPTION value per year (CON_YEAR + CON_VALUE) using the following formula: 

 
 

The resulting table will look as follows: 

The following table shows how the CONSUMPTION for INSTALLATION INST_002 is split across 2009 and 2010: 

 

 
 

SOLUTION 1 
Using SQL Joins 
Use traditional SQL processing. Three PROC SQL steps will be required. The execution of the first two queries 
involves performing a Cartesian product join that can not be optimized. 

Calculate the total SLP value for the consumption period. 

 

PROC SQL; 
   CREATE TABLE BC2_SOL1A AS 
      SELECT INSTALLATION, 
             MET_FROM, 
             MET_TO, 
             CONSUMPTION, 
             SUM (SLP) AS TOTAL_SLP 
         FROM METERING, 
              SLP_DAY 
         WHERE DATE BETWEEN MET_FROM AND MET_TO 
         GROUP BY INSTALLATION, 
                  MET_FROM, 
                  MET_TO, 
                  CONSUMPTION; 
QUIT; 

Calculate the total yearly SLP values. 

 

PROC SQL; 
   CREATE TABLE BC2_SOL1B AS 
      SELECT INSTALLATION,  
             YEAR (DATE) AS CON_YEAR, 
             MET_FROM, 
             MET_TO, 
             SUM (SLP) AS YEAR_SLP 
         FROM METERING, 
              SLP_DAY 
         WHERE DATE BETWEEN MET_FROM AND MET_TO 
         GROUP BY INSTALLATION, 
                  CALCULATED CON_YEAR, 
                  MET_FROM, 
                  MET_TO; 
QUIT; 

 

 
 

 
 

Split the CONSUMPTION value per year. 

 

PROC SQL; 
   CREATE TABLE BC2_SOL1 (DROP = TOTAL_SLP YEAR_SLP) AS 
      SELECT A.*, 
             CON_YEAR, 
             YEAR_SLP, 
             CONSUMPTION * YEAR_SLP / TOTAL_SLP AS CON_VALUE 
         FROM BC2_SOL1A A, 
              BC2_SOL1B B 
         WHERE A.INSTALLATION = B.INSTALLATION AND 
               A.MET_FROM = B.MET_FROM AND  
               A.MET_TO = B.MET_TO; 
QUIT; 

SOLUTION 2 
Using a Hash Table 
Load the SLP values into a hash table. 

 

DATA BC2_SOL2 (KEEP = INSTALLATION MET_FROM MET_TO CONSUMPTION CON_YEAR CON_VALUE); 
   SET METERING; 
   DROP RC; 
   IF _N_ = 1 THEN DO; 
      DECLARE HASH HT (DATASET : 'SLP_DAY');   
      HT.DEFINEKEY  ('DATE'); 
      HT.DEFINEDATA ('SLP'); 
      HT.DEFINEDONE (); 
   END; 
   TOTAL_SLP = 0; 
   DO DATE = MET_FROM TO MET_TO; 
      RC = HT.FIND ();   
      IF RC = 0 THEN TOTAL_SLP = TOTAL_SLP + SLP;   
   END; 
   CON_VALUE = 0; 
   YEAR_SLP = 0; 
   DO DATE = MET_FROM TO MET_TO; 
      RC = HT.FIND (); 
      IF RC = 0 THEN YEAR_SLP = YEAR_SLP + SLP;   
      IF DATE = MET_TO THEN DO; 
         CON_VALUE = YEAR_SLP / TOTAL_SLP * CONSUMPTION;   
         CON_YEAR = YEAR (DATE); 
         OUTPUT; 
      END; 
      ELSE DO; 
         IF YEAR (DATE) NE YEAR (DATE + 1) THEN DO; 
            CON_VALUE = YEAR_SLP / TOTAL_SLP * CONSUMPTION;   
            CON_YEAR = YEAR (DATE); 
            OUTPUT; 
            YEAR_SLP = 0; 
         END; 
      END; 
   END; 
RUN; 

 

 Load the SLP_DAY table into a hash object using the DATE variable as the key and the SLP values as the data. 
 Look up the SLP values for the consumption period in the hash object. 
 Calculate the total SLP value for the consumption period. 
 Calculate the total yearly SLP values. 
 Split the CONSUMPTION value per year. 

SOLUTION 3 
Using an Array 
Perform a table lookup in an array that contains the SLP values. 

 

PROC SQL NOPRINT;   
   SELECT MIN (DATE), 
          MAX (DATE) 
      INTO :ARR_START, 
           :ARR_STOP 
      FROM SLP_DAY; 
QUIT; 
 
%LET ARR_START = &ARR_START; 
%LET ARR_STOP  = &ARR_STOP; 
%LET ARR_RANGE = &ARR_START : &ARR_STOP; 
 
DATA BC2_SOL3 (KEEP = INSTALLATION MET_FROM MET_TO CONSUMPTION CON_YEAR CON_VALUE); 
   SET METERING; 
   FORMAT CON_VALUE COMMA12.2; 
   ARRAY SLP_VAL (&ARR_RANGE) SLP_VAL_&ARR_START - SLP_VAL_&ARR_STOP;   
   RETAIN SLP_VAL_&ARR_START - SLP_VAL_&ARR_STOP; 
   IF _N_ = 1 THEN DO; 
      DO INDEX1 = 1 TO NOBS; 
         SET SLP_DAY POINT = INDEX1 NOBS = NOBS; 
         SLP_VAL (DATE) = SLP;   
      END; 
   END; 
   TOTAL_SLP = 0; 
   DO INDEX2 = MET_FROM TO MET_TO; 
      TOTAL_SLP = TOTAL_SLP + SLP_VAL (INDEX2);   
   END; 
   CON_VALUE = 0; 
   YEAR_SLP = 0; 
   DO INDEX3 = MET_FROM TO MET_TO; 
      YEAR_SLP = YEAR_SLP + SLP_VAL (INDEX3);   
      IF INDEX3 = MET_TO THEN DO; 
         CON_VALUE = YEAR_SLP / TOTAL_SLP * CONSUMPTION;   
         CON_YEAR = YEAR (INDEX3); 
         OUTPUT; 
      END; 
      ELSE DO; 
         IF YEAR (INDEX3) NE YEAR (INDEX3 + 1) THEN DO; 
            CON_VALUE = YEAR_SLP / TOTAL_SLP * CONSUMPTION;   
            CON_YEAR = YEAR (INDEX3); 
            OUTPUT; 
            YEAR_SLP = 0; 
         END; 
      END; 
   END; 
RUN; 

 Create macro variables to specify the index range of the array. 
 

 
 

 Define the array to contain the daily SLP values. 
 

 
 

 Process the array to create the SLP_VAL variables containing the daily SLP values. 
 

 
 

 Calculate the total SLP value for the consumption period. 
 

 
 

 Calculate the total yearly SLP values. 
 Split the CONSUMPTION value per year. 

RESOURCE USAGE 
The following tables and charts illustrate resource usage in terms of: 

• real time 
• user CPU 
• memory 

BUSINESS CASE 3 
Combining Two Time-sliced Tables 

INPUT TABLES 
The CONTRACT table contains the period (CON_FROM – CON_TO) that an INSTALLATION is active. 

The METERING table contains the CONSUMPTION measured for an INSTALLATION between 2 dates (MET_FROM 
and MET_TO). 

We have to search for the periods in 2009 that although an INSTALLATION is active no CONSUMPTION has been 
measured. 

 

   
 

The following table shows the periods in 2009 that although an INSTALLATION is active no CONSUMPTION has 
been measured: 

 

 
 

The resulting table will look as follows: 

SOLUTION 1 
Using the DATA Step MERGE – BY Statement 
Use DATA step MERGE – BY statements to search for the periods in 2009 that although an INSTALLATION is active 
no CONSUMPTION has been measured. 

 

DATA _NULL_;   
   CALL SYMPUT ('START_RANGE', TRIM (LEFT ('01JAN2009'D))); 
   CALL SYMPUT ('END_RANGE', TRIM (LEFT ('31DEC2009'D))); 
RUN; 
 
PROC SORT DATA = CONTRACT; 
   BY INSTALLATION CON_FROM; 
RUN; 
 
DATA CON_DAY (KEEP = INSTALLATION DATE);   
   SET CONTRACT; 
   DO DATE = MAX (&START_RANGE, CON_FROM) TO MIN (&END_RANGE, CON_TO); 
      OUTPUT; 
   END; 
RUN; 
 
PROC SORT DATA = METERING; 
   BY INSTALLATION MET_FROM; 
RUN; 
 
DATA MET_DAY (KEEP = INSTALLATION DATE);   
   SET METERING;  
   DO DATE = MAX (&START_RANGE, MET_FROM) TO MIN (&END_RANGE, MET_TO); 
      OUTPUT; 
   END; 
RUN; 
 
DATA BC3_SOL1 (KEEP = INSTALLATION UNM_FROM UNM_TO); 
   MERGE CON_DAY (IN = C) MET_DAY (IN = M);   
   BY INSTALLATION DATE; 
   IF C AND NOT M;   
   RETAIN UNM_FROM PREV_DATE .; 
   FORMAT UNM_FROM UNM_TO DDMMYY10.; 
   IF FIRST.INSTALLATION THEN DO; 
      UNM_FROM = DATE; 
      PREV_DATE = DATE; 
   END; 
   IF DATE - PREV_DATE GT 1 THEN DO; 
      UNM_TO = PREV_DATE; 
      OUTPUT;   
      UNM_FROM = DATE; 
   END; 
   IF LAST.INSTALLATION THEN DO; 
      UNM_TO = DATE; 
      OUTPUT;   
   END; 
   PREV_DATE = DATE; 
RUN; 

 

 Create macro variables to restrict the analysis to 2009. 

 Split the CONTRACT table per day in 2009. 

 

 
 

 Split the METERING table per day in 2009. 

 

 
 

 Merge the CON_DAY and MET_DAY tables by INSTALLATION and DATE. 
 Only keep the observations from the CON_DAY table that do not exist in the MET_DAY table. 
 Output only one observation per period (UNM_FROM – UNM_TO). 

SOLUTION 2 
Using Arrays 
Use arrays to search for the periods in 2009 that although an INSTALLATION is active no CONSUMPTION has been 
measured. 

 

DATA _NULL_;   
   CALL SYMPUT ('START_RANGE', TRIM (LEFT ('01JAN2009'D))); 
   CALL SYMPUT ('END_RANGE', TRIM (LEFT ('31DEC2009'D))); 
RUN; 
 
DATA BC3_SOL2 (KEEP = INSTALLATION UNM_FROM UNM_TO); 
   MERGE CONTRACT (IN = C) METERING;   
   BY INSTALLATION; 
   IF C; 
   FORMAT UNM_FROM UNM_TO DDMMYY10.; 
   RETAIN CON_&START_RANGE - CON_&END_RANGE ' ' 
          MET_&START_RANGE - MET_&END_RANGE ' ' 
          UNM_&START_RANGE - UNM_&END_RANGE; 
   ARRAY CON (&START_RANGE : &END_RANGE) $ 1 CON_&START_RANGE - CON_&END_RANGE;   
   ARRAY MET (&START_RANGE : &END_RANGE) $ 1 MET_&START_RANGE - MET_&END_RANGE;   
   ARRAY UNM (&START_RANGE : &END_RANGE) $ 1 UNM_&START_RANGE - UNM_&END_RANGE;   
   IF FIRST.INSTALLATION THEN DO; 
      DO IR = &START_RANGE TO &END_RANGE; 
         CON (IR) = '-';   
         MET (IR) = '-';   
         UNM (IR) = '-';   
      END; 
   END; 
   DO IC = MAX (&START_RANGE, CON_FROM) TO MIN (&END_RANGE, CON_TO); 
      CON (IC) = 'X';   
   END; 
   DO IC = MAX (&START_RANGE, MET_FROM) TO MIN (&END_RANGE, MET_TO); 
      MET (IC) = 'X';   
   END; 
   IF LAST.INSTALLATION THEN DO; 
      DO IC = &START_RANGE TO &END_RANGE; 
         IF CON (IC) = 'X' AND MET (IC) NE 'X' THEN UNM (IC) = 'X';   
      END; 
      DO IC = &START_RANGE TO &END_RANGE; 
         IF UNM (IC) = 'X' THEN DO; 
            IF (IC NE &START_RANGE AND UNM (IC - 1) = '-') 
               OR IC = &START_RANGE THEN UNM_FROM = IC; 
            IF (IC NE &END_RANGE AND UNM (IC + 1) = '-') 
               OR (IC = &END_RANGE) THEN DO; 
               UNM_TO = IC; 
               OUTPUT;   
            END; 
         END; 
      END; 
   END; 
RUN; 

 Create macro variables to restrict the analysis to 2009. 

 

 
 

 Merge the CONTRACT and METERING tables by INSTALLATION. 

 

 
 

 
 

 Define the array CON based on the CONTRACT table to contain the days in 2009 that an INSTALLATION is 
active. 

 

 
 

 Define the array MET based on the METERING table to contain the days in 2009 that CONSUMPTION has been 
measured. 

 

 
 

 Define the array UNM to contain the days in 2009 that although an INSTALLATION is active no CONSUMPTION 
has been measured. 

 Process the CON, MET and UNM arrays to initialize the CON, MET and UNM variables to “-” for each 
INSTALLATION. 

 

 
 

 Process the CON array to assign the value “X” to the CON variables for the days in 2009 that an INSTALLATION 
is active. 

 Process the MET array to assign the value “X” to the MET variables for the days in 2009 that CONSUMPTION 
has been measured. 

 Process the CON, MET and UNM arrays to assign the value “X” to the UNM variables for the days in 2009 that 
although an INSTALLATION is active no CONSUMPTION has been measured. 

 

 
 

 Output only one observation per period (UNM_FROM – UNM_TO). 

SOLUTION 3 
Using Multiple SET Statements 
Use multiple SET statements to search for the periods in 2009 that although an INSTALLATION is active no 
CONSUMPTION has been measured. 

 

DATA _NULL_;   
   CALL SYMPUT ('START_RANGE', TRIM (LEFT ('01JAN2009'D))); 
   CALL SYMPUT ('END_RANGE', TRIM (LEFT ('31DEC2009'D))); 
RUN;  
 
DATA BC3_SOL3 (KEEP = INSTALLATION UNM_FROM UNM_TO); 
   SET CONTRACT;   
   FORMAT UNM_FROM UNM_TO DDMMYY10.; 
   RETAIN OBS 0; 
   CON_FROM = MAX (&START_RANGE, CON_FROM); 
   CON_TO = MIN (&END_RANGE, CON_TO); 
   FLAG = 0; 
   DO WHILE (FLAG = 0 AND OBS LT NOBS); 
      IF INSTALLATION GT MET_INSTALLATION 
         OR (INSTALLATION = MET_INSTALLATION AND CON_FROM GT MET_TO) THEN DO; 
         OBS = OBS + 1; 
         SET METERING (RENAME = (INSTALLATION = MET_INSTALLATION)) 
                      POINT = OBS NOBS = NOBS;   
         MET_FROM = MAX (&START_RANGE, MET_FROM); 
         MET_TO = MIN (&END_RANGE, MET_TO); 
      END; 
      IF MET_INSTALLATION GT INSTALLATION 
         OR (MET_INSTALLATION = INSTALLATION AND MET_FROM GT CON_FROM) THEN DO;   
         UNM_FROM = CON_FROM; 
         IF MET_INSTALLATION = INSTALLATION THEN UNM_TO = MIN (CON_TO, MET_FROM – 1); 
         ELSE UNM_TO = CON_TO; 
         OUTPUT; 
         IF MET_INSTALLATION GT INSTALLATION THEN FLAG = 1; 
      END; 
      IF FLAG = 0 AND INSTALLATION = MET_INSTALLATION THEN DO;   
         IF CON_TO GT MET_TO THEN DO; 
            IF CON_FROM LE MET_TO THEN CON_FROM = MET_TO + 1; 
         END; 
         ELSE FLAG = 1; 
      END; 
   END; 
   IF OBS = NOBS AND FLAG = 0 THEN DO;   
      IF INSTALLATION = MET_INSTALLATION THEN DO; 
         IF MET_FROM GT CON_FROM THEN DO; 
            UNM_FROM = CON_FROM; 
            UNM_TO = MET_FROM - 1; 
            OUTPUT; 
            CON_FROM = MET_FROM; 
         END; 
         IF MET_TO LT CON_TO THEN DO; 
            UNM_FROM = MET_TO + 1; 
            UNM_TO = CON_TO; 
            OUTPUT; 
         END; 
      END; 
      ELSE DO; 
         UNM_FROM = CON_FROM; 
         UNM_TO = CON_TO; 
         OUTPUT; 
      END; 
   END; 
RUN; 

 Create macro variables to restrict the analysis to 2009. 
 Read the CONTRACT table sequentially. 
 Read the METERING table using direct access. 
 Use complex conditional processing to search for days in 2009 that although an INSTALLATION is active no 

CONSUMPTION has been measured. 

 

RESOURCE USAGE 
The following tables and charts illustrate resource usage in terms of: 

• real time 
• user CPU 
• memory 

CONCLUSION 
You should consider spending more programmer time to reduce computer resource usage for complex combinations 
of tables with millions of records. So, instead of using the traditional programming techniques like SQL inner joins or 
DATA step MERGE – BY statements, you should consider using more complex programming techniques like hash 
tables, arrays and multiple SET statements. 

